

# NAG Library Routine Document

## G13AWF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G13AWF returns the (augmented) Dickey–Fuller unit root test.

### 2 Specification

```
FUNCTION G13AWF (TYPE, P, N, Y, IFAIL)
REAL (KIND=nag_wp) G13AWF
INTEGER TYPE, P, N, IFAIL
REAL (KIND=nag_wp) Y(N)
```

### 3 Description

If the root of the characteristic equation for a time series is one then that series is said to have a unit root. Such series are nonstationary. G13AWF returns one of three types of (augmented) Dickey–Fuller test statistic:  $\tau$ ,  $\tau_\mu$  or  $\tau_\tau$ , used to test for a unit root, a unit root with drift or a unit root with drift and a deterministic time trend, respectively.

To test whether a time series,  $y_t$ , for  $t = 1, 2, \dots, n$ , has a unit root, the regression model

$$\nabla y_t = \beta_1 y_{t-1} + \sum_{i=1}^{p-1} \delta_i \nabla y_{t-i} + \epsilon_t$$

is fitted and the test statistic  $\tau$  constructed as

$$\tau = \frac{\hat{\beta}_1}{\sigma_{11}}$$

where  $\nabla$  is the difference operator, with  $\nabla y_t = y_t - y_{t-1}$ , and where  $\hat{\beta}_1$  and  $\sigma_{11}$  are the least squares estimate and associated standard error for  $\beta_1$  respectively.

To test for a unit root with drift the regression model

$$\nabla y_t = \beta_1 y_{t-1} + \sum_{i=1}^{p-1} \delta_i \nabla y_{t-i} + \alpha + \epsilon_t$$

is fit and the test statistic  $\tau_\mu$  constructed as

$$\tau_\mu = \frac{\hat{\beta}_1}{\sigma_{11}}$$

To test for a unit root with drift and deterministic time trend the regression model

$$\nabla y_t = \beta_1 y_{t-1} + \sum_{i=1}^{p-1} \delta_i \nabla y_{t-i} + \alpha + \beta_2 t + \epsilon_t$$

is fit and the test statistic  $\tau_\tau$  constructed as

$$\tau_\tau = \frac{\hat{\beta}_1}{\sigma_{11}}$$

The distributions of the three test statistics;  $\tau$ ,  $\tau_\mu$  and  $\tau_\tau$ , are nonstandard. An associated probability can be obtained from G01EWF.

## 4 References

Dickey A D (1976) Estimation and hypothesis testing in nonstationary time series *PhD Thesis* Iowa State University, Ames, Iowa

Dickey A D and Fuller W A (1979) Distribution of the estimators for autoregressive time series with a unit root *J. Am. Stat. Assoc.* **74** 366 427–431

## 5 Arguments

- 1: TYPE – INTEGER *Input*  
*On entry:* the type of unit test for which the probability is required.  
 TYPE = 1  
 A unit root test will be performed and  $\tau$  returned.  
 TYPE = 2  
 A unit root test with drift will be performed and  $\tau_\mu$  returned.  
 TYPE = 3  
 A unit root test with drift and deterministic time trend will be performed and  $\tau_\tau$  returned.  
*Constraint:* TYPE = 1, 2 or 3.
- 2: P – INTEGER *Input*  
*On entry:*  $p$ , the degree of the autoregressive (AR) component of the Dickey–Fuller test statistic. When  $p > 1$  the test is usually referred to as the augmented Dickey–Fuller test.  
*Constraint:*  $P > 0$ .
- 3: N – INTEGER *Input*  
*On entry:*  $n$ , the length of the time series.  
*Constraints:*  
     if TYPE = 1,  $N > 2P$ ;  
     if TYPE = 2,  $N > 2P + 1$ ;  
     if TYPE = 3,  $N > 2P + 2$ .
- 4: Y(N) – REAL (KIND=nag\_wp) array *Input*  
*On entry:*  $y$ , the time series.
- 5: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, –1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by  $X04AAF$ ).

Errors or warnings detected by the routine:

$IFAIL = 11$

On entry,  $TYPE = \langle value \rangle$ .  
Constraint:  $TYPE = 1, 2$  or  $3$ .

$IFAIL = 21$

On entry,  $P = \langle value \rangle$ .  
Constraint:  $P > 0$ .

$IFAIL = 31$

On entry,  $N = \langle value \rangle$ .  
Constraint:  $N > \langle value \rangle$ .

$IFAIL = 41$

On entry, the design matrix used in the estimation of  $\beta_1$  is not of full rank, this is usually due to all elements of the series being virtually identical. The returned statistic is therefore not unique and likely to be meaningless.

$IFAIL = 42$

$\sigma_{11} = 0$ , therefore depending on the sign of  $\hat{\beta}_1$ , a large positive or negative value has been returned.

$IFAIL = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in *How to Use the NAG Library and its Documentation* for further information.

$IFAIL = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in *How to Use the NAG Library and its Documentation* for further information.

$IFAIL = -999$

Dynamic memory allocation failed.

See Section 3.7 in *How to Use the NAG Library and its Documentation* for further information.

## 7 Accuracy

None.

## 8 Parallelism and Performance

G13AWF is not threaded in any implementation.

## 9 Further Comments

None.

## 10 Example

In this example a Dickey–Fuller unit root test is applied to a time series related to the rate of the earth's rotation about its polar axis.

### 10.1 Program Text

```

Program g13awfe
!   G13AWF Example Program Text

!   Mark 26 Release. NAG Copyright 2016.

!   .. Use Statements ..
Use nag_library, Only: g01ewf, g13awf, nag_wp
!   .. Implicit None Statement ..
Implicit None
!   .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!   .. Local Scalars ..
Real (Kind=nag_wp)         :: pvalue, ts
Integer                     :: ifail, method, n, nsamp, p, type
!   .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: y(:)
Integer                     :: state(1)
!   .. Executable Statements ..

!   .. Executable Statements ..
Write (nout,*) 'G13AWF Example Program Results'
Write (nout,*)

!   Skip heading in data file
Read (nin,*)

!   Read in the problem size, test type, order of the AR process
Read (nin,*) n, type, p

!   Allocate memory
Allocate (y(n))

!   Read in the time series
Read (nin,*) y(1:n)

!   Calculate the Dickey-Fuller test statistic
ifail = 0
ts = g13awf(type,p,n,y,ifail)

!   Get the associated p-value using the look-up method
method = 1
ifail = -1
pvalue = g01ewf(method,type,n,ts,nsamp,state,ifail)

If (ifail==0 .Or. ifail==201) Then
!   Display the results
Write (nout,'(A,F6.3)') 'Dickey-Fuller test statistic   = ', ts
Write (nout,'(A,F6.3)') 'associated p-value           = ', pvalue
End If

End Program g13awfe

```

### 10.2 Program Data

```

G13AWF Example Program Data
30 1 1                                :: N,TYPE,P
-217 -177 -166 -136 -110 -95 -64 -37 -14 -25
-51 -62 -73 -88 -113 -120 -83 -33 -19 21
17 44 44 78 88 122 126 114 85 64 :: End of Y

```

### **10.3 Program Results**

G13AWF Example Program Results

Dickey-Fuller test statistic = -2.540  
associated p-value = 0.013

---